
Abstract. In applications of automata theory, one is interested in re-
ductions in the size of automata that preserve the recognised language.
For Büchi automata, two optimisations have been proposed: bisimulation
reduction, which computes equivalence classes of states and collapses
them, and α-balls reduction, which collapses strongly connected com-
ponents (SCCs) of an automaton that only contain one single letter as
edge label. In this paper, we present a formalisation of these algorithms
in Isabelle/HOL, providing a formally verified implementation.

1 Introduction

Model-checking is an important method for proving systems correct, and is ap-
plied in industrial practice [1]. In previous work [2], we present a reference im-
plementation for an LTL (linear temporal logic) model checker for finite-state
systems à la SPIN [5]. The model checker follows the well-known automata-
theoretic approach. Given a finite-state program P and an LTL formula φ, two
Büchi automata are constructed: the system automaton that recognises the ex-
ecutions of P , and the property or formula automaton expressing all potential
executions that violate φ, respectively. Then the product of the two automata is
computed and tested on-the-fly for emptiness. This implementation is realised
and verified using Isabelle/HOL [7].

One important part of automata-based model checking is the translation of an
LTL formula into a Büchi automaton. The standard algorithm for this problem
has been proposed by Gerth et al. [4]. Previously to [2], we have implemented
and verified this algorithm in Isabelle/HOL [8]. In this paper, we consider two
of the optimisations proposed by Etessami and Holzmann [3] to reduce the size
of the formula automaton.

In model checking, the system automaton is usually much larger than the
property automaton, but since the size of the property automaton is a mul-
tiplicative factor of the overall complexity, it is worthwhile to put substantial
effort into its optimisation [3].

The first optimisation is bisimulation reduction, which computes equivalence
classes of states and collapses them. The algorithm of [3] uses a so-called colouring.

⋆ Supported by DFG grant CAVA, Computer Aided Verification of Automata.



1: proc BasicBisimReduction(A) ≡
2: /* Init: ∀q ∈ Q. C−1(q) := 1, and ∀q ∈ F. C0(q) := 1, ∀q ∈ Q \ F. C0(q) := 2.*/
3: i := 0;
4: while |Ci(Q)| �= |Ci−1(Q)| do
5: i := i+ 1
6: foreach q ∈ Q do

7: Ci(q) := 〈Ci−1(q),∪(q,a,q′)∈δ{(C
i−1(q′), a)}〉

8: od

9: Rename colour set Cn(Q), with {1, . . . , |Ci(Q)|}, using lexicogr. ordering.
10: od

11: C := Ci; return A′ := 〈Q′ := C(Q), δ′, q′I := C(qI), F
′ := C(F )〉;

12: /* δ′ defined so that (C(q1), a, C(q2)) ∈ δ′ iff (q1, a, q2) ∈ δ for q1, q2 ∈ Q*/

Fig. 1. Basic Bisimulation Reduction Algorithm [3]

Our formalisation has revealed that there is a mistake in the initialisation of the
algorithm, which we have corrected in our implementation.

The second optimisation is α-balls reduction, which collapses strongly con-
nected components (SCCs) that only contain one single letter as edge label.

The rest of the paper is organised as follows: Section 2 gives some preliminar-
ies. Section 3 recalls the two optimisations of [3] in turn. Section 4 presents our
Isabelle formalisations of those algorithms, and Sec. 5 concludes.

2 Preliminaries

We recall the basic notions of automata as used by [3]; for more details see [10].
Usually, Büchi (or finite) automata have transitions labelled with characters

from an alphabet Σ. In [3], a generalisation of such labellings is considered, but
for our purposes, this is not necessary and so we assume simple characters. We
assume that a Büchi automaton A is given by 〈Q, δ, qI , F 〉. Here Q is a set of
states, δ ⊆ (Q×Σ×Q) is the transition relation, qI ∈ Q is the initial state, and
F ⊆ Q is the set of final states. The language L(A) is defined as the set of those
ω-words which have an accepting run in A, where a run on word w = a1a2 . . . is
a sequence q0q1q2 . . . such that q0 = qI and (qi, ai+1, qi+1) ∈ δ for all i ≥ 0, and
it is accepting if qi ∈ F for infinitely many i.

Isabelle/HOL [7] is an interactive theorem prover based on Higher-Order Logic
(HOL). You can think of HOL as a combination of a functional programming
language with logic. Isabelle/HOL aims at being readable by humans and thus
follows the usual mathematical notation, but still the syntax needs some expla-
nations which we provide when we come to the examples. In our presentation of
Isabelle code we have stayed faithful to the sources.

3 The Original Algorithms

3.1 The Bisimulation Reduction Algorithm

Fig. 1 shows the basic bisimulation reduction algorithm in pseudo-code. The let-
ter C with a superscript refers to the iterations of the computation of a colouring.



Fig. 2. An automaton and its incorrect reduction according to [3]

The idea is that in the beginning (i = 0) accepting states have colour 1 and non-
accepting states have colour 2, and in each step, the colour of a state is obtained
by its old colour and a combination of the successor state colours and the cor-
responding edge labels. This means that if two states have the same colour but
they differ in the colours of their successors (taking into account the edge labels),
then those two states must be distinguished; we say that the colouring is refined.
In the end, states with the same colour can be joined.

The algorithm initialises not only C0 but also C−1 (we might call this “pre-
initialisation”) which is a trick making the formulation of the algorithm more
concise, by allowing for a loop condition that makes a comparison between the
current and the previous colouring, even for i = 0.

However, our formalisation of the reduction algorithm, to be shown later, has
revealed that there is a mistake in this pre-initialisation. This is illustrated in
Figure 2. Here we have Q = {q0, q1}, F = Q,C0(q0) = 1, C0(q1) = 1, C−1(q0) =
1, C−1(q1) = 1. Just before the while we have |C0(Q)| = |C−1(Q)| = 1; even
stronger, we have C0(Q) = C−1(Q) = {1}. What matters is that the loop
condition is false and hence the loop is not entered at all. Therefore the result
C = C0 is computed, yielding the automaton shown on the right in Figure 2.

Our explanation is as follows: The pre-initialisation ∀q ∈ Q. C−1(q) := 1 is
conceptually wrong. It expresses that “pre-initially” (i = −1), there is only one
colour. If by coincidence the input automaton has only accepting or only non-
accepting states, then “initially” (index i = 0), there is also just one colour. The
loop condition will then wrongly say “we have done enough refinement steps”.

The problem really manifests itself for the case that F = Q, i.e., all states are
accepting: each refinement step takes into account the edge labels and not just
whether a state is accepting or not. The initialisation however only considers
whether a state is accepting or not, and so not doing any refinement wrongly
results in identifying all states (q0 and q1 in the example).

The conceptual mistake happens to cause no harm in the case F = ∅, since
the initialisation of the algorithm establishes the property F = ∅ and trivially
maintains it since no refinement is done. The accepted language is then empty.

Our solution is to replace the condition |Ci(Q)| 	= |Ci−1(Q)| with i ≤ 0 ∨
|Ci(Q)| 	= |Ci−1(Q)| (in the actual formalisation: i > 0 −→ |Ci(Q)| 	= |Ci−1(Q)|)



so that the loop body will definitely be entered for i = 0 at least once. This is
shown in Fig. 3 and will be discussed in Sec. 4.1.

3.2 α-Balls Reduction

This optimisation may appear simple and rather specialised, but in fact, it is
quite effective in our context of model checking, more precisely, on Büchi au-
tomata that are the result of a translation from generalised Büchi automata
which in turn are the output of the formula translation [4]. Note also that the
reduction does not work for finite automata; it only works for Büchi automata.

The idea of the reduction is that, if in a Büchi automaton we are ever stuck
in a component, and the only transition labels in this component are α, and
there is some accepting state in the component, then we can treat the entire
component as a single accepting state with a self-transition labelled by α.

Definition 1. For α ∈ Σ, a fixed-letter α-ball1 inside a Büchi automaton A is

a set Q′ ⊆ Q of states such that:

1. α ∈ Σ is the unique letter which labels the transitions inside Q′;

2. the nodes of Q′ form an SCC of A;
3. there is no transition leaving Q′ , i.e., no (q′, b, q) ∈ δ where q′ ∈ Q′ and

q /∈ Q′.

4. Q′ ∩ F 	= ∅.

Proposition 1. Given a Büchi automaton A = 〈Q, δ, qI , f〉, suppose Q′ ⊆ Q is

a fixed-letter α-ball of A. Let A′ = 〈(Q \Q′) ∪ {qnew}, δ′, q′I , (F \Q′) ∪ {qnew}〉
where

δ = {(q1, b, q2) | q1, q2 ∈ Q \Q′} ∪ {(q1, b, qnew ) | (q1, b, q2) ∈ δ, q1 ∈ Q, q2 ∈ Q′}
∪ {(qnew , α, qnew )},

and q′I = qnew if qI ∈ Q′, else q′I = qI . Then L(A) = L(A′).

4 Isabelle Formalisation

The work presented in this paper is a fragment of a bigger library being developed
on automata in the context of model checking, in particular the construction of
the property automaton (see Sec. 1). Modularity, generality and reuse are impor-
tant concerns in this project, which is why the Isabelle code chunks presented
here exhibit some aspects that we do not discuss in all detail.

Generally, automata are represented as record types that are parametrised by
the type of the states and the type of the alphabet, among others. E.g., in line 2
in Fig. 3, ′q is the type of the states. The fields of these records, mostly denoted
by calligraphic letters, refer to the states, the final states, etc. E.g., in line 5 in
Fig. 3, Q gives the state set, and in line 7, F refers to being accepting.

[3] defines more generally: a fixed-formula α-ball, i.e., α is a formula.



1: definition LBA_bpr_C ::

2: "(’q, ’l, ’more) LBA_scheme ⇒ (’q ⇒ nat) nres"

3: where

4: "LBA_bpr_C A ≡ do {

5: let Q = Q A;

6: let C = (λq. 1);

7: let C’ = (λq. if F A q then 1 else 2);

8: let i = 0;

9: (_, C’, _) ←
10: WHILEIT

11: (LBA.LBA_bpr__whilei A Q)

12: (λ(C, C’, i). i>0 −→ card (C’ ‘ Q) �= card (C ‘ Q))

13: (λ(C, C’, i). do {

14: let i = Suc i;

15: let C = C’;

16: let f = (λq. (C q, L A q, C ‘ successorsA A q));

17: let R = f ‘ Q;

18: fi ← set_enum R;

19: let C’ = fi o f;

20: RETURN (C, C’, i)

21: }) (C, C’, i);

22: RETURN C’

23: }"

Fig. 3. BPR colouring

In the formalisation we present here, we use automata that follow [4] but
differ from the standard definition in one important aspect: we assume that
not the edges, but rather the states of automata are labelled. We call those
automata labelled Büchi automata (LBA), as opposed to BA. Moreover, in our
representation we have a set of initial states rather than a unique initial state.
Instead of a set of accepting states we use a predicate to express whether a state is
an accepting state or not. This kind of automata representation is suitable in our
context, since we formalise the algorithm in the context of the Büchi automaton
construction from an LTL formula according to [4], where the output automaton
corresponds to a state labelled rather than a transition labelled automaton.

Big parts of our library concern BAs, however, and technically, LBAs are
defined as an extension of the BA record type where labels for the states are
added and the labels for the edges are “disabled”.

4.1 The Bisimulation Reduction Algorithm

The Isabelle formalisation of the algorithm from Fig. 1 is shown in Figure 3.
The formalisation uses the Isabelle refinement framework [6] for writing what
resembles imperative code. We explain some of the syntactic constructs.

Lines 5 to 8 accomplish the initialisation. The let should be understood as
imperative assignment. We assign:Q is the state set of the input automatonA; C



1: definition

2: "set_enum S ≡ do {

3: (_, m) ← FOREACHi

4: (set_enum__foreachi S)

5: S

6: (λx (k, m). RETURN (Suc k, m(x �→k)))

7: (1, empty);

8: RETURN (λx. case m x of None ⇒ 0 | Some k ⇒ k) }"

Fig. 4. Numbering of sets

is the (i−1)th colouring and is initialised to the function that colours each state
as 1; C′ is the ith colouring and is initialised to the function that colours each
accepting state as 1 and all others as 2. We need C and C′ in order to determine
whether the current iteration has actually refined the current colouring.

Line 9 performs a kind of nondeterministic assignment: the term following it
is essentially a set of values, and an element of this set is assigned to ( , C′, )
nondeterministically. The refinement framework allows us to specify algorithms
with such nondeterminism, prove theorems about them, and replace the nonde-
terminism by a deterministic implementation later and independently.

Back to the code: line 10 contains the while-construct in this language. Its
first argument (line 11) consists of a loop invariant that must be provided by the
programmer and that is used in correctness proofs. The second argument (line
12) is the loop condition corrected as explained in Sec. 3.1. The third argument
(lines 13 to 21) is the loop body which takes the form of a λ-term with an
abstraction over (C,C′, i), applied to the argument (C,C′, i) (line 21) which
corresponds to the initial values explained two paragraphs above.

During each iteration, a new colouring is computed in the form of a function
f that assigns a certain triple to each state; R is then defined as the image of f
on Q, i.e., R is the set of all the colours of the new colouring. In order to convert
those complicated triples into simple numbers, an auxiliary function shown in
Fig. 4 is used; fi , obtained by a nondeterministic assignment as in line 9, is then
the numbering of R. The new colouring is then the composition of fi and f . It
assigns a number to each state.

The function set enum computes the set of all unique numberings for a set
S, so that set enum S is the set of bijections between S and {1, . . . ,#S}. The
definition of the function starts with a foreach-construct (lines 3-7). In line 3
one possible result of the loop is chosen non-deterministically and is assigned
to ( ,m). In the following line a loop-invariant is provided in order to be able
to prove correctness properties of the definition. The second parameter in line 5
represents the iterated set. For each element of S the body of the loop (line 6) is
applied sequentially in an arbitrary order, where x is an element of S and (k,m)
an intermediate result of the loop, that is propagated through the iterations
starting with (1, empty). Such results correspond to a pair consisting of the next
number to assign and an already constructed mapping of numbers to elements



of S. An empty mapping is denoted by empty and for an existing mapping m
a new mapping is constructed with m(x �→ k), that behaves like m with the
exception that it maps x to k. In the last line the above constructed mapping is
turned into a function: elements not in S are mapped to 0. The following lemma
states the correctness of the construction:

lemma set_enum_correct:

assumes "finite S"

shows "set_enum S ≤ SPEC (λf. bij_betw f S {1..card S})"

The term bij betw f S {1..card S} says “f is a bijection between S and
{1, . . . ,#S}”, and the entire lemma says that the results of set enum S in
terms of functions f fulfil the specification “f is a bijection between S and
{1, . . . ,#S}”.

Thus we have a bijection between f(Q) and C′(Q) in the actual procedure,
i.e., line 9 in the pseudo-code in Fig. 1 is correctly implemented.

Termination After each execution of the body of the loop, the number of colours
according to C′ increases compared to C, or the iteration stops after that execu-
tion of that loop body. At the same time, the number of possible colours in C′ is
bounded by the number of states in the input automaton. Hence the number of
colours cannot increase indefinitely and therefore the iteration stops eventually.

Correctness The proof of correctness of the procedure is based on the charac-
terisation of the ith colouring. For i > 0, we have the following loop invariant:

definition ( in LBA)

"LBA_bpr_C_inv Q ≡ λ(C, C’, i).

∀ q∈Q. ∀ q’∈Q. C’ q = C’ q’

←→ (C q = C q’

∧ L A q = L A q’

∧ C ‘ successors q = C ‘ successors q’)"

This invariant corresponds exactly to the modification of C′ after each iteration
and is thus simple to prove based on the bijectivity of f and C′. For the general
case, i.e., including i = 0, the following holds: whenever C′(q) = C′(q′) for two
states q, q′ ∈ Q, then either both q, q′ are in F or they are both not in F .

When the iteration of the loop stops, i.e., the number of colours in C′ does
not change anymore compared to C, we obtain a bijection between C and C′.
Considering that the invariant relates C to C′, we obtain that C and C′ yield the
same equivalence classes, i.e., for any states q, q′ ∈ Q, we have C′(q) = C′(q′) if
and only if C(q) = C(q′). In Fig. 5, we have defined the characteristics that are
needed in the subsequent proofs and that LBA bpr C indeed fulfils. We have
chosen to introduce a definition for this characterisation because it occurs in
several places in our development.

All in all, we obtain the characterisation of the resulting colouring shown
by the lemma in Figure 6. Indeed, we need to assume that the set of states is



definition ( in LBA)

"LBA_bpr_C_char

≡ λC. ∀ q∈Q A. ∀ q’∈Q A.

C q = C q’ −→ (F A q ←→ F A q’)

∧ L A q = L A q’

∧ C ‘ successors q = C ‘ successors q’"

Fig. 5. Characterisation of the colouring

lemma ( in LBA) LBA_bpr_C_correct:

"LBA_bpr_C A ≤ SPEC LBA_bpr_C_char"

Fig. 6. Correctness property of the colouring

finite. Otherwise we could not apply the above shown properties about set enum.
Finiteness is given here by an implicit assumption denoted by “(in LBA)”.

This characterisation turns out to be a sufficient condition for proving the
correctness of the resulting coloured automaton. To obtain the automaton, we
apply a renaming function LBA rename which takes an LBA and a colouring
function C and returns the LBA where each state has been renamed using the
colouring function. The renaming function has properties shown in Figure 7.

For example the second line of the lemma states that the initial states of the
renamed automaton are exactly the renamings of the initial states of the original
automaton. The other lines state similarly that the transition function, the final
states, and the labels are preserved by the renaming.

The term inv into Q f q gives “the”2 inverse element of q under f in Q. In
the case that this inverse is unique, i.e., f is injective on Q, it is straightforward
to show that the renaming preserves language equivalence. However, the very
purpose of the colouring is to reduce the number of states, hence not to be injec-
tive! But even in this general case, language equivalence holds, as is expressed
by the following lemma:

lemma ( in LBA) LBA_bpr_C_rename_accept_iff:

assumes "LBA_bpr_C_char C"

shows "∀ w. LBA_accept (LBA_rename A C) w ←→ LBA_accept A w"

The proof of the lemma works constructively. The “←−” direction consists
of taking a run (sequence of states) r for word w of the input automaton and
colouring r componentwise using C, i.e., r is mapped to a run C ◦ r in the
coloured automaton.

The “−→” direction requires an auxiliary function:

2 This is the famous ǫ operator of HOL: It represents an arbitrary but fixed term
fulfilling a given property.



lemma LBA_rename_simps:

"δ (LBA_rename A C) q a = C ‘ δ A (inv_into (Q A) C q) a"

"I (LBA_rename A C) = C ‘ I A"

"F (LBA_rename A C) q ←→ F A (inv_into (Q A) C q)"

"L (LBA_rename A C) q = L A (inv_into (Q A) C q)"

Fig. 7. Renaming function for LBAs

q1
a

q2
a

1
a

Fig. 8. An LBA and its colouring

fun bpr_run

where

"bpr_run A C r q0 0 = q0"

| "bpr_run A C r q0 (Suc k)

= (SOME q’. C q’ = C (inv_into (Q A) C (r (Suc k)))

∧ q’∈δL A (bpr_run A C r q0 k))"

This function is needed for the following reason: if we start from a run r for w
in the coloured automaton and use inv into Q f q to compute a corresponding
state in the input automaton for each state q in r, then these states do not
necessarily “fit together”, i.e., they may not form a run in the input automaton.

Example 1. Figure 8 shows an LBA accepting the word aaa . . . on the left, and
the simplified automaton obtained by colouring on the right. Obviously, the
state sequence in the coloured automaton is always simply a sequence of 1’s.
The inverse of “1” is either q1 or q2, so simply translating 11 . . . back into the
original LBA would give either q1q1 . . . or q2q2 . . .. In neither case this would
correspond to a run of the original LBA.

As the example shows, constructing a run in the original LBA, given a run
in the coloured LBA, is not so simple and has to be done step by step starting
from the initial state. This is what the function bpr run is good for. It starts by
taking an inverse of the initial state of the run r of the coloured automaton, and
then always picks an appropriate inverse for each next state in r. The existence
of such an inverse is guaranteed by the assumption that the colouring C fulfils
the characterisation according to Figure 5.

The entire procedure for constructing the coloured automaton is then given
in Figure 9. By the considerations given above, this construction is correct:

lemma ( in LBA) LBA_bpr_correct:

"LBA_bpr A
≤ SPEC (λAC. LBA AC ∧ (∀ w. LBA_accept AC w ←→ LBA_accept A w))"



definition LBA_bpr :: "(’q, ’l, _) LBA_scheme ⇒ (nat, ’l) LBA nres"

where

"LBA_bpr A ≡ do {

C ← LBA_bpr_C A;

RETURN (LBA_rename A C)

}"

Fig. 9. Procedure for constructing the coloured automaton

In addition to language equivalence we also show that the coloured automaton
is well-formed, i.e. LBA AC holds.

4.2 α-Balls Reduction

Most of the Isabelle development on this topic uses in fact BA, not LBA, and the
formalisation of α-balls reduction for LBA is based on the one for BA. However,
we focus here on LBA because the use of LBA required certain adaptations that
partly make the contribution of our work.

An α-ball for LBA has, of course, an Isabelle definition. However, this is not
very readable and so we prefer to present the following characterisation:

lemma αballL_full_def:

"αballL A α Q ≡
G A ↾ Q ∈ sccs A ∧ Q �= {}

∧ (∀ q∈Q. successors A q ⊆ Q)

∧ (∀ q∈Q. L A q = α)"

The lemma says that Q is an α-ball iff the following four conditions hold: (1)
the graph of A, restricted to Q, is an SCC of A; (2) Q 	= ∅; (3) there is no edge
leading out of Q; (4) all states in Q are labelled with α.

Fig. 10 gives an auxiliary definition for balls reduction. There is a loop for
working through the SCCs. Trivial balls or balls without accepting states are
removed. One-element balls are left untouched. Finally, non-trivial balls are col-
lapsed to a single state. Based on the computation of SCCs according to Tarjan’s
algorithm [9] we obtain a function that computes the SCCs and then reduces,
as shown in Figure 11. The ball reduction on LBAs thus returns a well-formed
equivalent automaton and a set SCCsR representing its SCCs.

The correctness follows from the correctness of LBA reduce ball aux and
tarjan :

lemma ( in LBA) LBA_reduce_balls_correct:

"LBA_reduce_balls A ≤ SPEC(λ(AR, SCCsR).

LBA AR ∧ (∀ w. LBA_accept AR w ←→ LBA_accept A w)

∧ Q AR =
⋃
SCCsR

∧ pairwise_disjoint SCCsR

∧ (∀ Q∈SCCsR. G AR ↾ Q∈sccs AR ∧ Q �= {})

)"



definition LBA_reduce_balls_aux ::

"[(’q, ’l, ’more) LBA_scheme, ’q set set]

⇒ ((’q, ’l, ’more) LBA_scheme × ’q set set) nres"

where

"LBA_reduce_balls_aux A SCCs

≡ FOREACHi

(LBA_reduce_balls_aux_foreach_inv A SCCs)

SCCs

(λQ (AR, SCCsR).

if ¬ ballL AR Q then RETURN (AR, {Q} ∪ SCCsR)

else if scc_trivial AR (G AR ↾ Q)

then RETURN (remove_states AR Q, SCCsR)

else if ∀ q∈Q. ¬ F AR q

then RETURN (remove_states AR Q, SCCsR)

else if (∃ q. Q = {q}) then RETURN (AR, {Q} ∪ SCCsR)

else

case ballL_get_α AR Q of

None ⇒ RETURN (AR, {Q} ∪ SCCsR)

| Some α ⇒
do { (AR, QR) ← LBA_remove_αball AR Q;

ASSERT (QR⊆Q ∧ QR �= {});

RETURN (AR, {QR} ∪ SCCsR) }

)

(A, {})"

Fig. 10. Auxiliary definition for α-balls reduction

definition LBA_reduce_balls ::

"(’q, ’l, ’more) LBA_scheme

⇒ ((’q, ’l, ’more) LBA_scheme × ’q set set) nres"

where

"LBA_reduce_balls A ≡ do {

SCCs ← tarjan A (I A);

(AR, SCCsR) ← LBA_reduce_balls_aux A SCCs;

RETURN (AR, SCCsR)

}"

Fig. 11. Ball reduction on LBAs

5 Conclusion

We have presented an Isabelle/HOL formalisation of two Büchi automata opti-
misations proposed by [3]. The context of this work is explained in detail in [2]:
implementing full-fledged model checkers verified in Isabelle/HOL. Within this
endeavour, it is worthwhile to invest effort in the optimisation of the property
automaton. While the optimisations are particularly relevant for model checking,
they are abstract enough to be applicable to Büchi automata in general.



The difficulty lay in making the right design decisions for the formalisation.
We generally tried to model the Isabelle proofs on the paper-and-pencil proofs
of the literature, but especially for the colouring, it turned out to be better
to develop a new constructive proof from scratch. In that particular case, as
mentioned above, the original code contained a mistake, which we discovered
during our vain efforts to prove the code correct.

Concerning the first optimisation, there is also a more sophisticated algorithm
[3]. However this algorithm relies on the edge labels in an automaton and thus
it is not immediately possible to implement this algorithm in our scenario.

Our formalisation consists of around 1500 lines of code (approximately 3000
lines of code if the formalisation of Tarjan’s SCC algorithm is counted).

